Transistor characterization

Worksheet for computations

■ Some useful functions

In[1]:=
```
LSolve = Last[Solve[##]] &
```
Out[1]=
```
Last[Solve[##1]] &
```
In[2]:=
```
PeS = PowerExpand[Simplify[##]] &
FpPeS = FixedPoint[PeS, ##] &
```
Out[2]=
```
PowerExpand[Simplify[##1]] &
```
Out[3]=
```
FixedPoint[PeS, ##1] &
```
In[4]:=
```
NDCollect = Function[{expr, s, func}, Collect[Numerator[expr], s, func] / Collect[Denominator[expr], s, func]]
```
Out[4]=
```
Function[{expr, s, func}, Collect[Numerator[expr], s, func] / Collect[Denominator[expr], s, func]]
```

Small signal analysis

Feedback VCVS connected between V_{gate} and V_{drain}.

■ Small signal MOSFET and V_{gate} feedback equations

I_d: drain current
V_g: gate voltage
V_d: drain voltage
g_m: transconductance
g_o: small signal output conductance
V_{dref}: reference drain voltage
A : Voltage gain for $V_g$feedback regulation
v_n: gate referred MOSFET noise voltage

In[5]:=
```
eqn = {
   Id == gm (Vg + vn) + go Vd,
   Vg == Vd + A (Vd - Vdref)
  }
```
Out[5]=
```
{Id == go Vd + gm (Vg + vn), Vg == Vd + A (Vd - Vdref)}
```

■ Solve for V_{gate} and V_{drain}

In[6]:=
```
sol = FpPeS[LSolve[eqn, {Vg, Vd}]]
```
Out[6]=
$$\left\{Vg \to \frac{Id + A\,Id - A\,go\,Vdref - gm\,vn - A\,gm\,vn}{gm + A\,gm + go},\ Vd \to \frac{Id + A\,gm\,Vdref - gm\,vn}{gm + A\,gm + go}\right\}$$

Substitute gain $A \equiv 1 / \alpha$ so that infinite gain is $\alpha = 0$

In[7]:=
```
FpPeS[sol /. A → α^-1]
% /. α → 0
```

Out[7]= $\left\{Vg \to \frac{-go\ Vdref + Id\ (1+\alpha) - gm\ vn\ (1+\alpha)}{gm + gm\,\alpha + go\,\alpha},\ Vd \to \frac{gm\ Vdref + Id\,\alpha - gm\ vn\,\alpha}{gm + gm\,\alpha + go\,\alpha}\right\}$

Out[8]= $\left\{Vg \to \frac{Id - go\ Vdref - gm\ vn}{gm},\ Vd \to Vdref\right\}$

- V_{gate}

Split into terms dependent on I_d, V_{dref}, and v_n

In[9]:=
```
Collect[Vg /. sol, {Id, Vdref, vn}, Simplify]
```

Out[9]= $\frac{(1+A)\ Id}{gm + A\ gm + go} - \frac{A\ go\ Vdref}{gm + A\ gm + go} - \frac{(1+A)\ gm\ vn}{gm + A\ gm + go}$

Substitute $A \equiv 1/\alpha$ and simplify for infinite gain

In[10]:=
```
NDCollect[#, α, Simplify] & /@ Collect[Vg /. sol /. A → α^-1, {Id, Vdref, vn}, Simplify]
% /. α → 0
```

Out[10]= $-\frac{go\ Vdref}{gm + (gm + go)\,\alpha} + \frac{Id + Id\,\alpha}{gm + (gm + go)\,\alpha} + \frac{-gm\ vn - gm\ vn\,\alpha}{gm + (gm + go)\,\alpha}$

Out[11]= $\frac{Id}{gm} - \frac{go\ Vdref}{gm} - vn$

- **I_d sweep**
- $\frac{\partial V_g}{\partial I_d}$

In[12]:=
```
D[Vg /. sol, Id]
```

Out[12]= $\frac{1 + A}{gm + A\ gm + go}$

- **V_{dref} sweep**
- $\frac{\partial V_g}{\partial V_{\text{dref}}}$

In[13]:=
```
D[Vg /. sol, Vdref]
```

Out[13]= $-\frac{A\ go}{gm + A\ gm + go}$

- $\frac{\partial V_d}{\partial V_{\text{dref}}}$

In[14]:=
```
D[Vd /. sol, Vdref]
```

Out[14]= $\frac{A\ gm}{gm + A\ gm + go}$

- **Implicit differentiation** $-\frac{\partial V_d}{\partial V_{\text{dref}}} \Big/ \frac{\partial V_g}{\partial V_{\text{dref}}}$

In[15]:=
```
- D[Vd /. sol, Vdref] / D[Vg /. sol, Vdref]
```

Out[15]= $\frac{gm}{go}$

■ Observed gate voltage noise with feedback

■ $\frac{\partial V_g}{\partial v_n}$

In[16]:=
```
Simplify[D[Vg /. sol /. go → γ gm /. LSolve[1 + A == α^-1, A], vn]] /. γ → go / gm
% /. LSolve[1 + A == α^-1, α]
Vg == %% vn
Simplify[LSolve[%, vn] /. go → γ gm] /. γ → go / gm /. LSolve[1 + A == α^-1, α]
```

Out[16]= $\frac{1}{-1-\frac{\text{go}\,\alpha}{\text{gm}}}$

Out[17]= $\frac{1}{-1-\frac{\text{go}}{(1+A)\,\text{gm}}}$

Out[18]= $\text{Vg} == \frac{\text{vn}}{-1-\frac{\text{go}\,\alpha}{\text{gm}}}$

Out[19]= $\left\{\text{vn}\to-\left(1+\frac{\text{go}}{(1+A)\,\text{gm}}\right)\text{Vg}\right\}$